
      !   RE-STORE "Corel_wp.prg"
      !
      !***********************************************************************************
      !  "WordPerfect example" for Tech Soft's MS Office function library
      !
      !  (c) Tech Soft GmbH 1999-2006                 http://www.TechSoft.de
      !  Author:           Sven Henze
      !  Date:             03-Sep-2001
      !  Last change:      02-May-2006
      !
      !***********************************************************************************
      !
      ! Please note: You must set the variable Wpwinvers$ to match your version of WordPerfect !
      !
      INTEGER Verbose=1           ! switch on/off verbose mode (0=off)
      !
      ! Load the high level function library if not already loaded
      IF NOT INMEM("Dde_lib_start") THEN LOADSUB ALL FROM "htbdde.lib"
      !
      !
      ! Now we call the custom SUB where the Office functions will be executed
      ! If you need more calling parameters either include them in the SUB definition
      ! or use COM blocks
      !
      CALL Dde_mainprog(Verbose)
      !
      !
      ! cleaning up SUB programs (in order to make sure that the most recent versions are loaded every time)
      DELSUB Dde_lib_start TO END
      !
      END
      !
      SUB Dde_mainprog(OPTIONAL INTEGER Verbose)
      !
      ! ***********************************************************************
      ! *                                                                     *
      ! *  MS Office Interface for HTBasic: Corel WordPerfect example         *
      ! *                                                                     *
      ! *  Version :  2.01.42                                                 *
      ! *  Author  :  Sven Henze, Tech Soft GmbH                              *
      ! *  Created :  03-Sep-2001                                             *
      ! *  Updated :  02-May-2006                                             *
      ! *                                                                     *
      ! *  Copyright (c) 1999-2006  Tech Soft GmbH. All Rights Reserved.      *
      ! *                           Email: HTB@TechSoft.de                    *
      ! *                           http://www.techsoft.de/dde.htm            *
      ! *                                                                     *
      ! ***********************************************************************
      !
      ! Description:
      ! ============
      !
      ! This program show how to insert text and images into a WordPerfect document.
      ! Please be aware that it may take several seconds to load WordPerfect and to
      ! connect to the template file. This is a limitation of the WordPerfect system.
      !
      ! You must set the variable Wpwinvers$ to match your version of WordPerfect !
      !
      ! Please note:
      !   This program was written and optimized for HTBasic 8.3 and higher.
      !   If you use an older version of HTBasic it might be neccessary to remove
      !   or change some commands. The example does *not* work with HTBasic 7.x due
      !   to some problems in the CSUB interface within HTBasic 7.x.
      !
      ! define some variables
          DIM Doc_template$(1:10)[255],Doc_file$[255],App_exe$[255],Topic$[255],Tmp$[8192],Htbvers$[80],Os$[255],Question$[1024]
          INTEGER Printer,Btn,Sys_chn(1:5),Doc_chn(1:10),Result(1:10),App_quit,Language,Printlen,Dummy
          DIM Ddecmd$[255],Info$[2048],Bookmark$[128],Doc_password$[128],Sel$[255],Struct_file$[255]
          DIM Top$[4096],Sname$[255],Value$[1024],App_name$[255],App$[255],Text$[8192],Ddedriver$[255],Img_file$[255]
          DIM G$[5],X$[5],Hl$[5],Hx$[5],Crlf$[8],Wpwinvers$[20],Lf$[5],Tb$[5],Newfile$[5]
          !
          Crlf$=CHR$(10)&CHR$(13)
          !
          ON ERROR GOSUB Dde_err_handler        ! Global error handler
          !
          G$=CHR$(19)                                  ! character code for inverse printing
          X$=CHR$(16)                                  ! character code for normal printing
          Hl$=CHR$(30)                                 ! character code for "highlighted" printing (different color)
          Hx$=CHR$(24)                                 ! character code for "normal" printing (color)
          !
          INTEGER Verb_
          Verb_=0
          IF NPAR>=1 THEN
              IF Verbose>0 THEN Verb_=1
          END IF
          !
          Printlen=FNDde_defprintlen  ! define printlen for this session
          !
          ! define colors and other settings
          INTEGER Custom_colours
          Custom_colours=1                       ! use custom colours (set to 0 if you do not wish to change colour palette)
          !
          IF NOT FNDde_init_system(Custom_colours) THEN STOP
          !
          ! Application quit behaviour
          ! Values:   0  = Do not close the application ever
          !           1  = Close the application after all DDE commands have been sent without any warning
          !           2  = Prompt the user to close the application
          !          >2  = App_quit=Timeout value (prompt the user to close the application until the given timeout has reached)
          !                If timeout occurs the dialog will be closed and the application will not be closed automatically
          App_quit=2
          !
          App$="WPWIN"     ! Corel WordPerfect.

          !
          ! IMPORTANT !!!  You must select the proper version number for Wpwinvers$ !!
          !
          !  Valid numbers are: "7", "8", "9", "10", "11", "12", "13"
         ! Wpwinvers$="12"           ! use this if you are using WordPerfect 12
         ! Wpwinvers$="13"          ! use this if you are using WordPerfect X3 (release 13)
          !
          ! you can remove the following line in your application if you properly set Wpwinvers$ before
          GOSUB Select_wpvers        ! select WordPerfect version with Basic Plus dialog
          !
          App_name$="WordPerfect"
          Topic$=App$&Wpwinvers$
          App_exe$=Topic$&".exe"
          !
          !
          ! If the specified filename starts with a "*" a new file will be created and saved
          !Newfile$="*"              ! uncomment this if you want to create a new file
          !
          IF Newfile$="*" THEN
              ! If the specified filename starts with a "*" a new file will be created and saved
              Doc_template$(1)=Newfile$&SYSTEM$("MSI")&"\newfile.wpd"   ! Create new empty document
          ELSE
              Doc_template$(1)=SYSTEM$("MSI")&"\template.wpd"           ! Open WordPerfect template
          END IF
          Doc_template$(2)=SYSTEM$("MSI")&"\template-output.wpd"        ! WordPerfect output file
          !
          IF Verb_>0 THEN CALL Dde_programhead("HTBasic - Corel WordPerfect example")
          !
          CALL Dde_start(App_exe$,Topic$,Sys_chn(1),Language,Verb_)      ! start the DDE application defined in Topic$
          !
          IF NOT Sys_chn(1) THEN
              BEEP
              GOTO Finish
          END IF
          !
          ALPHA PEN 3
          IF Verb_>0 THEN OUTPUT CRT;" "&CHR$(20)&"Sending DDE commands to "&App_name$&X$
          ALPHA PEN 1
          !
          !
          ! first open a document in WordPerfect
          Doc_chn(1)=FNDde_openfile(Topic$,Sys_chn(1),Doc_template$(1),Result(1),Doc_password$,Verb_)
          !
          IF NOT Doc_chn(1) THEN
              BEEP
              GOTO Finish
          END IF
          !
          !
 Myown_dde_cmds: !
          !
          !==========================================================================================================
          !
          !  ******  Custom DDE programming starts here  ******
          !
          ! Todo: Write your own Office programming code here
          !
          Lf$=CHR$(10)
          Tb$=CHR$(9)
          !
        !  CALL Dde_driver_info        ! shows info about DDE driver version and date
          !
          REAL Fre_mem,Divisor
          Divisor=1024
          Fre_mem=FRE/(Divisor)
          Fre_mem=PROUND(Fre_mem/1024,-1)
          !
          ! fill the text$ array
          Text$="MS Office Interface Test for WordPerfect"&Lf$&Lf$
          Text$=Text$&"Welcome to the MS Office Interface test for Corel WordPerfect."&Lf$
          Text$=Text$&"This text was transfered from BASIC for Windows using the MS Office Interface extension from Tech Soft."&Lf$
          Text$=Text$&"A demo version of the MS Office Interface and more info can be found here:"&Lf$
          Text$=Text$&"       http://www.TechSoft.de/mso"&Lf$&Lf$&Lf$
          Text$=Text$&"****** System information ******"&Lf$
          Text$=Text$&"Date:"&Tb$&Tb$&Tb$&DATE$(TIMEDATE)&Lf$
          Text$=Text$&"Time:"&Tb$&Tb$&Tb$&TIME$(TIMEDATE)&Lf$
          Text$=Text$&"BASIC version:"&Tb$&SYSTEM$("VERSION:HPBW")&Lf$
          Text$=Text$&"MSI:"&Tb$&Tb$&Tb$&SYSTEM$("MSI")&Lf$
          Text$=Text$&"Available memory:"&Tb$&VAL$(Fre_mem)&" MB"&Lf$
          Text$=Text$&"Htbdde.dll Driver:  "&Tb$&FNDde_driver_vers$&Lf$
          Text$=Text$&"Htbdde.lib Driver:  "&Tb$&FNDde_lib_version$&Lf$&Lf$
          !
          ! goto start of document
          Ddeexecute(Doc_chn(1),"PosDocTop",Result(1))
          !
          ! insert text in WordPerfect document
          Text$="Type(Text:"&CHR$(34)&Text$&CHR$(34)&")"
          Dde_fprint("  -> Inserting text ",Printlen-2,Verb_)
          Ddeexecute(Doc_chn(1),Text$,Result(1))
          Dde_show_status(Result(1),Verb_)
          Dde_fprint("  -> Formatting text ",Printlen-2,Verb_)
          Ddeexecute(Doc_chn(1),"PosDocTop",Result(1))
          !
          ! select entire line
          Ddeexecute(Doc_chn(1),"SelectLineEnd",Result(1))
          !
          ! change text attribute to "Bold"
          Ddeexecute(Doc_chn(1),"AttributeAppearanceToggle(Attrib: Bold!)",Result(1))
          Ddeexecute(Doc_chn(1),"PosDocTop",Result(1))
          Dde_show_status(Result(1),Verb_)
          !
          Ddeexecute(Doc_chn(1),"PosDocBottom",Result(1))  ! go to the end of the document
          !

          ! now we insert a picture file
          Img_file$=SYSTEM$("MSI")&"\"&"specplot.gif"                              ! desired image file
          Dde_fprint("  -> Inserting picture '"&Hl$&Img_file$&"'"&Hx$,Printlen,Verb_)
          Ddeexecute(Doc_chn(1),"BoxCreate (BoxType:""I&mage"")",Result(1))        ! insert box
          Ddeexecute(Doc_chn(1),"BoxContentType (Content:Image!)",Result(1))       ! make it an image box
          Ddeexecute(Doc_chn(1),"BoxImageRetrieve (Action: MakeInternal!; Filename: """&Img_file$&""" )",Result(1))   ! specify ext
ernal filename
          Ddeexecute(Doc_chn(1),"BoxHorizontalAlignment (Alignment: AlignColumns!; Position: FullAlign!; Offset: 5)",Result(1))  ! 
align picture
          Ddeexecute(Doc_chn(1),"BoxBorder (Style: ButtonBorder!)",Result(1))      ! apply border
          Ddeexecute(Doc_chn(1),"BoxUpdateDisplay ()",Result(1))                   ! update display
          Ddeexecute(Doc_chn(1),"BoxEnd (State: Save!)",Result(1))                 ! save edits
          Dde_show_status(Result(1),Verb_)
          !
          !
          !  ******  Custom DDE programming ends here  ******
          !
          !==========================================================================================================
          !
          CALL Dde_focus2basic                       ! bring HTBasic window back to front
          !
          !
          IF Newfile$="*" THEN
              ! if the file was newly created simply save it under the given name
              Dummy=FNDde_savefile(Topic$,Doc_chn(1),Doc_template$(1),Result(1),2,"",Verb_)
          ELSE
              ! Save file under the name specified in Doc_template$(2). Ask for overwrites
              Dummy=FNDde_savefile(Topic$,Doc_chn(1),Doc_template$(2),Result(1),0,"",Verb_)
          END IF
          !
          Dde_enter_kbd(" Press <RETURN> to close the document(s).")
          !
          ! Close file and ask if file should be saved
          ! Doc_chn(1)=FNDde_closefile(Topic$,Sys_chn(1),Doc_template$(1),Result(1))
          !
          ! Close file, do not ask for save
          Dummy=FNDde_closefile(Topic$,Sys_chn(1),Doc_template$(1+Dummy),Result(1),1,Verb_)
          !
 Finish:  !
          ! terminate all DDE connections and resources
          ! if variable "App_quit" is set to 1 or greater the application will be closed
          CALL Dde_quit(Sys_chn(1),Doc_chn(1),Result(1),App_quit,Topic$,Verb_,0,1)
          !
          ALPHA PEN 4
          DISP " Program finished."
          ALPHA PEN 1
          !
          SUBEXIT
          !
          ! Global error handler
 Dde_err_handler:  !
          BEEP
          Dde_enter_kbd("**** Global Error Handler ****"&Crlf$&Crlf$&ERRM$)
          ERROR RETURN
          !
 Select_wpvers: ! select WordPerfect version with Basic Plus dialog
          ! subprogram returns the WordPerfect version number in variable Wpwinvers$
          DIM M$(1:7)[15],P$[80],But$(1:1)[50]
          INTEGER Sel,I
          P$="Select your WordPerfect version from"&Crlf$
          P$=P$&"the list below and press 'Continue'."
          But$(1)="Continue"
          !
          FOR I=7 TO 13
              M$(I-6)="WordPerfect "&VAL$(I)
              IF I=13 THEN M$(I-6)="WordPerfect X3"
          NEXT I
          !
          DIALOG "COMBO",P$,Btn;SET ("TITLE":" WordPerfect Version ","DIALOG BUTTONS":But$(*),"WIDTH":40,"ITEMS":M$(*),"EDITABLE":0,
"SELECTION":6,"JUSTIFICATION":"LEFT"),RETURN ("SELECTION":Sel)
          !
          Wpwinvers$=VAL$(Sel+7)       ! return the selected version number
          RETURN
          !
      SUBEND


